
This section discusses the relationships between NonStop NET/MASTER
Management Services (MS) users, NonStop NET/MASTER MS environments,
NonStop NET/MASTER MS commands, and NCL processes.

The section begins by discussing the types of users that can use NonStop
NET/MASTER MS and the types of environments in which users and NCL processes
operate.

The concept of environments is closely related to how NonStop NET/MASTER MS
commands are executed from an NCL process.  Accordingly, the section also discusses
the two ways to enter NonStop NET/MASTER MS commands from an NCL process:
by using the CMD core statement and by using the INTCMD verb.  In particular, it
discusses where CMD and INTCMD send the results of NonStop NET/MASTER MS
commands.

Then the section discusses the two queues associated with a dependent processing
environment:  the request queue and the response queue.  Finally, the section
discusses how to send a message to an NCL process by using the INTQ command, and
how to send a message between NCL processes by using the WRITE verb.

Table 16-1 summarizes the core statement, verb, and NonStop NET/MASTER MS
commands used to initiate processing in an environment.

Table 16-1.  Core Statement, Verb, and NonStop NET/MASTER MS Commands Used to Initiate
Processing in an Environment

Core Statement, Verb,
or Command Description

CMD core statement Issues a NonStop NET/MASTER MS command for execution in the current
execution environment.

EXEC command Invokes an NCL procedure for serial execution.

INTCMD verb Executes a NonStop NET/MASTER MS command in the dependent
processing environment of an NCL process.

START command Invokes an NCL procedure for asynchronous execution.

START verb Invokes an NCL procedure for asynchronous execution in the specified
environment using the specified variables.

Table 16-2 summarizes the verbs and NonStop NET/MASTER MS command used to
handle messages in an environment.

Table 16-2.  Verbs and NonStop NET/MASTER MS Command Used to Handle Messages in an
Environment

Verb or Command Description

INTCLEAR verb Discards outstanding messages queued to a dependent processing
environment.

INTCONT verb Propagates a message read using the INTREAD verb to the next higher
processing environment and, optionally, changes its message text.

INTQ command Passes data to the dependent request or response queue associated with
an NCL process.

INTREAD verb Retrieves the next message from the dependent request or response queue
associated with an NCL process.

WRITE verb Writes a message to a terminal, an NCL process, all monitor-class users, a
specified user, or the activity log.

For the complete syntax of core statements and verbs, refer to the NonStop
NET/MASTER NCL Reference Manual.  For the complete syntax of NonStop
NET/MASTER MS commands, refer to the NonStop NET/MASTER MS Command
Reference Manual.

Note This section has many figures to illustrate the concepts discussed.  In the figures:

“NCL” refers to an NCL process.

“NNM” refers to a NonStop NET/MASTER MS command.

“REQ” refers to a request queue.

“RESP” refers to a response queue.

Types of NonStop
NET/MASTER MS

Users

There are three types of users that can use a NonStop NET/MASTER MS system, each
of which is discussed in the following subsections:

Real users

Virtual users

Remote users

Real Users A real user is a person who logs on to a NonStop NET/MASTER MS system at a local
terminal.  This type of user operates in a region (discussed later in this section) and has
a NonStop NET/MASTER MS profile, created by a user ID definition record, that
determines what the user can do in their region.

The NonStop NET/MASTER MS profile of a real user may allow the user to enter
Operator Control Services (OCS).  If so, the user has an operator profile and can use
one or two OCS windows.  Associated with each window is a primary processing
environment (defined later in this section) in which NonStop NET/MASTER MS
commands and NCL processes execute.

A typical example of a real user is a system operator who controls NonStop
NET/MASTER MS at a local system, performing tasks such as monitoring local users
and terminals, by using NonStop NET/MASTER MS commands from OCS.

Changing the Profile of a Real User

The profile of a real user is changed using User ID Management Services (UMS).  This
changes the details in the user ID definition record of the user.  If the details are
changed, the user can use the SIGNON command from OCS (while logged on to
NonStop NET/MASTER MS) to implement the changes and refresh their profile.  The
following command refreshes your own profile after your user ID definition record is
changed:

SIGNON

From OCS, a real user can use the PROFILE command to display or modify some
aspects of their operator profile.  The following command displays your own operator
profile:

PROFILE

 Caution If a user is not authorized to receive EMS, PPO, or monitor-class messages, the PROFILE command
does not display information about the EMS or MONMSG operand.  An NCL process can attempt to
check whether a user is allowed to receive these messages by using the INTCMD PROFILE statement
and waiting for message numbers NNM0378 (EMS operand) and NNM0366 (MONMSG operand).  If the
user is not authorized to receive these messages, the message numbers will never arrive and the NCL
process may be suspended indefinitely.  You can work around this problem by using the SECCALL GET
or SECCALL QUERY verb to help you obtain the actual message receipt status of the user.

The following command changes your own operator profile to allow the receipt of
unsolicited messages:

PROFILE UNSOL=YES

The SHOW USERS and SHOW OCS commands display information about real users.

Virtual Users Virtual users are NonStop NET/MASTER MS users that act the same as a real user
and have the same attributes as real users operating in OCS, except that they do not
have an associated terminal.  Virtual users, however, have a logical terminal and a
logical window that logically operates in OCS and, therefore, has a primary processing
environment associated with it.

Every NonStop NET/MASTER MS system has, by default, five system-wide virtual
users:

The Background Logger (BLOG)

The Background Monitor (BMON)

The Background System Process (BSYS)

The Event Management Service Process (EMSP)

The Activity Log Process (LOGP)

These virtual users are created by default when NonStop NET/MASTER MS starts,
and exist as long as NonStop NET/MASTER MS is running (unless you explicitly stop
them).  Since virtual users do not have a terminal associated with them, they are
regarded as working in a background processing environment.

Table 16-3 summarizes the main job of the virtual users.

Table 16-3.  Virtual Users

Virtual User Job

Background Logger (BLOG) Executes NonStop NET/MASTER MS commands
and logs the results.

Background Monitor (BMON) Executes NonStop NET/MASTER MS commands
and sends results to monitor-class users.

Background System Process (BSYS) Executes the INIT and READY NCL procedures
during NonStop NET/MASTER MS startup.

Event Management Service Process (EMSP) Controls the EMSPROC NCL procedure.

Activity Log Process (LOGP) Controls the LOGPROC NCL procedure.

Changing the Profile of a Virtual User

The profile of a virtual user is changed using User ID Management Services (UMS).
This changes the details in the user ID definition record of the user.  If the details are
changed, a real user can use the SUBMIT and SIGNON commands from OCS to
implement the changes and refresh the profile of the virtual user.  The following
command refreshes the profile of the Background Monitor after its user ID definition
record is changed:

SUBMIT BMON SIGNON

From OCS, a real user can use the SUBMIT and PROFILE commands to display or
modify some aspects of the profile of a virtual user.  The following command displays
the profile of the Background Monitor:

SUBMIT BMON PROFILE

You must be able to receive both monitor-class and unsolicited messages to see the
results of this command.  The following command allows you to do this:

PROFILE UNSOL=YES MONMSG=YES

Settings of the PROFILE command that determine command entry and retention in
OCS, message display on an OCS window, and the appearance of an OCS window are
not meaningful to a virtual user.  Therefore, the operands of the PROFILE command
that determine these settings are not included in the results of the command.

The following command changes the operator profile of the Background Monitor to
allow the receipt of Event Management Service (EMS) and unsolicited messages:

SUBMIT BMON PROFILE EMS=YES UNSOL=YES

Note If the user ID definition record of the Background Monitor does not allow it to receive EMS messages, the
EMS=YES operand is ignored.

The SHOW USERS command displays information about virtual users.

Note You can automate message processing by a virtual user by specifying a MSGPROC NCL procedure for
the user and by changing its user ID definition record to enable the receipt of EMS, monitor-class, and
unsolicited messages (by changing the values in the Receive EMS Messages, Monitor Status, and
Unsolicited Message Receipt fields, respectively, in UMS).

Remote Users A remote user is a person who logs on to a remote NonStop NET/MASTER MS
system (using the SIGNON link-name command) after logging on to a NonStop
NET/MASTER MS system at a local terminal.

This type of user has at least two NonStop NET/MASTER MS profiles.  One profile is
created by a user ID definition record at the local system and determines what the user
can do at the local system.  Another profile is created by a user ID definition record at
the remote system and determines what the user can do at the remote system.

From the point of view of the local system, this type of user is a real user or virtual
user who is allowed to enter OCS and so has an OCS profile.  From the point of view
of the remote system, this user is a type of virtual user (because the user does not have
an associated terminal at the remote system).

A typical example of a remote user is a network operator who controls many NonStop
NET/MASTER MS systems from a local system.  The network operator logs on to each
remote system and uses the ROUTE command from the local system to send
commands to each remote system.

Changing the Profile of a Remote User

The profile of a remote user is changed using User ID Management Services (UMS) at
both the local and remote system.  This changes the details in the user ID definition
records of the user.

If the details are changed at the local system, the user can use the SIGNON command
from OCS (while logged on to NonStop NET/MASTER MS at the local system) to
implement the changes and refresh their local profile.  The following command
refreshes your local profile at your local system after your local user ID definition
record is changed:

SIGNON

If the details are changed at the remote system, the user can use the ROUTE and
SIGNON commands from OCS (while logged on to NonStop NET/MASTER MS at the
remote system) to implement the changes and refresh their remote profile.  The
following command refreshes your remote profile at the remote system, REMOTE1,
after your remote user ID definition record is changed:

ROUTE REMOTE1 SIGNON

From OCS, a local user can use the PROFILE command to display or modify some
aspects of their local operator profile at the local system.  The following command
displays your local operator profile:

PROFILE

From OCS, a remote user can use the ROUTE and PROFILE commands to display or
modify some aspects of their remote operator profile at the remote system.  The
following command displays your remote operator profile:

ROUTE REMOTE1 PROFILE

The following command changes an operator profile at the local system to allow the
receipt of unsolicited messages:

PROFILE UNSOL=YES

The following command changes an operator profile at the remote system to allow the
receipt of unsolicited messages:

ROUTE REMOTE1 PROFILE UNSOL=YES

You can use the SHOW USERS and SHOW OCS commands to display information
about remote users.

Note You can filter messages from a remote system by specifying a MSGPROC NCL procedure at either the
local system, the remote system, or both systems.  The following command specifies a MSGPROC at
your local system:

PROFILE MSGPROC=LOCFILT

The following command specifies a MSGPROC at the remote system to which you are logged on:

ROUTE REMOTE1 PROFILE MSGPROC=RMTFILT

Types of NonStop
NET/MASTER MS

Environments

To NonStop NET/MASTER MS, an environment is an area in which NonStop
NET/MASTER MS users, NonStop NET/MASTER MS commands, and NCL
processes operate.  An environment provides the internal services and facilities that
are required to execute NonStop NET/MASTER MS commands and NCL processes on
behalf of NonStop NET/MASTER MS users.

Conceptually, an environment defines the scope of the area in which users, NonStop
NET/MASTER MS commands, and NCL processes operate.  It is designed to limit the
scope of the default activities of NonStop NET/MASTER MS commands and NCL
processes.  The concept of an environment provides a boundary between:

Multiple users and their activities
Multiple NCL processes and their activities
Users and NCL processes and their activities

The following environmental concepts are discussed in this subsection:

Regions
Current execution environments
Windows and top-level environments
Primary processing environments
Background processing environments
Dependent processing environments

Note An NCL process can determine the environment in which it is executing by using the &SYS.NCL.ENV
system variable.

Regions A region is an environment in which a real user operates.  It defines the scope of what
a real user can do when the user is logged on to NonStop NET/MASTER MS.
Conceptually, a region is designed to limit the default activities of real users.  This has
two consequences:

NonStop NET/MASTER MS commands issued within a region affect only the user
that owns the environment.

This means that you may require a relatively high authority level to use a NonStop
NET/MASTER MS command that affects another user.  An example is the

NonStop NET/MASTER MS SHUTDOWN command.  This command affects all
users and requires an authority level of at least 4.

NCL processes executed within a region affect only the user that owns the
environment.

This prevents one user from affecting or tampering with NCL processes in use by
another user.  The settings of system parameters, such as SYSPARMS NCLXUSER,
affects the ability of some NonStop NET/MASTER MS commands and command
operands to cross a region boundary.  An example is the INTQ command,
described later in this section.

Environmental Scope and Hierarchy

Figure 16-1 shows the regions of two real users, NNMJPN and NNMDGS, in which the
former is logged on twice.  The activities of each user are clearly bound by their
regions.  Available activities include choosing options from the primary menu,
displaying panels, and executing NonStop NET/MASTER MS commands and NCL
procedures.

Figure 16-1.  Regions of Two Real Users

Region 1: NNMJPN

Primary menu, NCL processes, NNM commands, Panels • • •

Region 2: NNMDGS

Primary menu, NCL processes, NNM commands, Panels • • •

Region 3: NNMJPN

Primary menu, NCL processes, NNM commands, Panels • • •

011

Figure 16-2 illustrates the concept of scope in more detail.  It shows that, within each
region, there are two top-level environments and that, below each top-level
environment, there is one primary processing environment.  Each primary processing
environment is associated with an OCS window (OCS window 1 and OCS window 2).
NonStop NET/MASTER MS commands and NCL procedures executed from an OCS
window run in the primary processing environment associated with that window.

The OCS command input line prompt is => (and can be preceded by mode indicators,
such as A, P, and M:  for example, APM=>).  You can execute any NonStop

NET/MASTER MS command (if you have sufficient authority) from the OCS
command input line, including the EXEC and START commands.  These commands
execute NCL procedures from OCS.

Figure 16-2 shows that the EXEC command queues NCL procedures for execution
(because only one NCL process executed by EXEC can execute at a time in a
processing environment).  It shows that the START command executes an NCL
procedure independently of other NCL processes (because NCL processes executed by
the START command can execute simultaneously in an environment).  The difference
between the EXEC and START commands is explained in detail in Section 8,
“Executing NCL Procedures.”

Figure 16-2.  Scope of a Region

Region

OCS Window

==> NNM command
       (including EXEC and 
START)

Top-Level Environment 1 

Primary Processing Environment

NCL process (executed with START)

NCL process (executed with EXEC)

012

NNM command

OCS Window 1 =>

Environments are created in a hierarchy.  Figure 16-3 shows the full environmental
hierarchy for a real user, beginning with a region.  The figure shows that every NCL
process has a dependent processing environment associated with it, and that the
hierarchy of that environment is theoretically unlimited.  (There is, by default, a
maximum of 128 NCL processes that a user can execute concurrently.  This is set by
the SYSPARMS NCLUMAX command.)  Figure 16-3 also shows that every dependent
processing environment that is created operates within the scope of the region.
Compare Figure 16-3 with Figure 16-6, later in this section.

Figure 16-3.  Environmental Hierarchy for a Real User

Real User
(With a Terminal)

013

Top-Level Environment
(NNM Primary Menu Window 1)

Primary Processing Environment

NCL NNM NCL NNM

Dep. P.E. Dep. P.E.

NCL NNM NCL NNM

Dep. P.E. Dep. P.E.

OCS Window 1

Primary Processing Environment

NNM NCL NNM NCL

Dep. P.E. Dep. P.E.

NNM NCL NNM NCL

Dep. P.E. Dep. P.E.

OCS Window 2

Top-Level Environment
(NNM Primary Menu Window 2)

Region

Current Execution
Environments

A current execution environment is the environment in which a NonStop
NET/MASTER MS command or NCL process is executing.  It refers to the
environment from the point of view of that command or NCL process.  A current
execution environment is one of the following:

A top-level environment
A primary processing environment
A background processing environment
A dependent processing environment

Input/Output and Current Execution Environments

The concept of a current execution environment is important because the current
execution environment determines where the output of a command executed in that
environment is sent after processing.

For example, if you enter the SHOW OCS command from the OCS command input
line, the command is executed in the primary processing environment associated with
the OCS window.  The results of the command are sent to your OCS window and
displayed in the roll-delete message display area.

However, if the virtual user BLOG, which executes in a background processing
environment, executes the SHOW OCS command, the command is executed in its
primary processing environment, and the results of the command are sent only to the
activity log.

Finally, if an NCL process issues the SHOW OCS command for execution in its
dependent processing environment, the results are sent to the NCL process for
additional processing.

Table 16-4 summarizes how the environment in which a NonStop NET/MASTER MS
command is executed affects where the results of the command are sent.  It also shows
the owner of each type of current execution environment.

Table 16-4.  Input/Output and Current Execution Environments

Environment Owner of Environment Sends Results to

Top-level Real user Screen.

Primary OCS window OCS window of real user in local system.

Background Remote user Environment of remote user in remote system.

Background BLOG Activity log.

Background BMON Terminals of monitor-class users and the activity log.
When displayed, messages are preceded by the M
monitor prefix.

Background BSYS Activity log.

Background EMSP Terminals of monitor-class users and the activity log.
When displayed, messages are preceded by the E
monitor prefix.

Background LOGP Terminals of monitor-class users and the activity log.
When displayed, messages are preceded by the L
monitor prefix.

Dependent NCL process Controlling NCL process, which determines what to do
with the results.

Windows and Top-Level Environments

NonStop NET/MASTER MS has two windows available for each user who logs on.

Each window has a number (1 or 2), each window is referred to by its number
(window 1 or window 2), and each window keeps its number while a user is logged
on, regardless of the activity that takes place.

Window 1 is created automatically when a user logs on:  by default, when a user logs
on,  NonStop NET/MASTER MS displays the NonStop NET/MASTER MS primary
menu in window 1.  Window 2, however, is not created automatically.  The window is
created on the first occasion a user swaps from window 1 or splits the screen from
window 1:  by default, when created, the NonStop NET/MASTER MS primary menu
is displayed in window 2.

Note A window is created by being displayed on the screen.  A window that has been displayed at least once
on the screen is called an open window.  A closed window is a window that has not been displayed at all
on the screen.  However, an open window may not always be visible on the screen:  for example, when
using two full-screen windows and swapping between them, only one window can be displayed at a time.

A top-level environment is the environment in which the NCL processes that display
NonStop NET/MASTER MS menus operate.  These are NCL processes that display
the panels that show the NonStop NET/MASTER MS primary menu, the User ID
Management Services (UMS) menus, the Edit Services selection list, the System
Support Services menus, and so on.

Every real user, who is allowed to choose options from a NonStop NET/MASTER MS
menu, has a top-level environment, and every real user owns its top-level
environment.

Primary Processing Environments

If a NonStop NET/MASTER MS command or NCL process is executing in a primary
processing environment, the current execution environment is the primary processing
environment.

Every real and virtual user has a primary processing environment, and every user
owns its primary processing environment.

A real user executes a NonStop NET/MASTER MS command (including the START
and EXEC commands) in a primary processing environment by typing it at the OCS
command input line.  The command is then sent for command processing.  A virtual
user executes a NonStop NET/MASTER MS command in a primary processing
environment by simply sending it for command processing.

If an NCL process, which is executed by a real user using either the START or EXEC
command from the OCS command input line or by a virtual user, executes a NonStop
NET/MASTER MS command using the CMD core statement (including the START
and EXEC commands), the NonStop NET/MASTER MS command executes in a
primary processing environment.  (The START and EXEC commands and the CMD
core statement are discussed later in this section.)

The output from a NonStop NET/MASTER MS command executed in a primary
processing environment owned by an OCS window goes to that window.  The output
from a NonStop NET/MASTER MS command executed in a primary processing
environment owned by a virtual user goes to the background processing environment
in which the virtual user operates.

Figure 16-4 shows the flow of NonStop NET/MASTER MS commands typed at the
OCS command input line.

Figure 16-4.  Flow of Commands and Results From OCS

Command Processing

 ==> NNM Command

Primary Processing Environment

OCS Window
(Message Area)

Results

Figure 16-5 shows the flow of NonStop NET/MASTER MS commands executed using
the CMD core statement from an NCL process that is running in a primary processing
environment.

Figure 16-5.  Flow of Commands and Results From NCL Under OCS

==> START/EXEC

Primary Processing Environment

OCS Window
(Message Area)

Results

NCL
CMD NNM

CMD EXEC/
         START

NCL
CMD NNM

CMD EXEC/
         START

NCL
CMD NNM

CMD EXEC/
         START

ResultsResults

Note A real user may have a MSGPROC NCL procedure that intercepts and processes messages flowing to an
OCS window.  This procedure can choose not to pass messages to the window.

Background Processing Environments

If a NonStop NET/MASTER MS command or NCL process is executing in a
background processing environment, the current execution environment is the
background processing environment.

A background processing environment is an environment that is owned and
controlled by a virtual user.  You can use the NonStop NET/MASTER MS SUBMIT
command to send a NonStop NET/MASTER MS command or NCL procedure to a
virtual user to execute on your behalf.  You can also use the KEEP and ROUTE
operands of the AT and EVERY timer commands to send a timer command to a virtual
user to execute.

Figure 16-6 shows the full environmental hierarchy for a virtual user.  The difference
between the hierarchy for a virtual user and the hierarchy for a real user is that a
virtual user does not have a top-level environment and has only one primary
processing environment.

Figure 16-6.  Environmental Hierarchy for a Virtual User

Primary Processing Environment

Background Processing Environment

Virtual User
(Does Not Have a Terminal)

NCL NNM NCL NNM

Dep. P.E. Dep. P.E.

NCL NNM NCL NNM

Dep. P.E. Dep. P.E.

The output of NonStop NET/MASTER MS commands executed in a background
processing environment is processed in environment-dependent ways:  what happens
to the output depends on the virtual user that controls the background processing
environment.

For example, if the Background Logger (BLOG) controls the environment, the results
are sent only the activity log.  If the Background Monitor (BMON) controls the
environment, the results are sent to all monitor terminals, as well as the activity log.
Table 16-4, earlier in this section, summarizes where results from NonStop
NET/MASTER MS commands are sent when executed by a virtual user.

Dependent Processing Environments

If a NonStop NET/MASTER MS command or NCL process is executing in a
dependent processing environment, the current execution environment is the
dependent processing environment.

A dependent processing environment is an environment that is controlled and owned
by an NCL process.  Every NCL process can have a dependent processing
environment associated with it.  This means that there can be a hierarchy of dependent
processing environments.  (The hierarchy is illustrated in Figure 16-6, earlier in this
section.)

A dependent processing environment inherits the profile of its higher-level
environment (except for the value of the UNSOL operand of the PROFILE command,
which is always initially equal to NO).  You can determine the profile of the
environment by using the PROFILE command.

Settings of the PROFILE command that determine command entry and retention in
OCS, message display on an OCS window, and the appearance of an OCS window are
not meaningful to a dependent processing environment.  Therefore, the operands of
the PROFILE command that determine these settings are not included in the results of
the command.

You can use the INTCMD verb from an NCL process to send a NonStop
NET/MASTER MS command to its dependent processing environment for execution.
The NCL process can then either read the results, using the INTREAD verb, or discard
the results, using the INTCLEAR verb.  After it reads the results, it can send the results
to the next higher processing environment using the INTCONT verb.  The INTCMD,
INTREAD, INTCLEAR, and INTCONT verbs are discussed later in this section.

The continuing execution of NonStop NET/MASTER MS commands and NCL
processes in a dependent processing environment depends upon the continuing
execution of the NCL process that controls the environment.  If an NCL process
terminates, all NonStop NET/MASTER MS commands and NCL processes that are
executing in the dependent processing environment—and all others below that—
terminate as well.

Note When you exit from OCS, all NonStop NET/MASTER MS commands executing in the window and NCL
processes executed from the window terminate.  You can use the FLUSH command to immediately
terminate an NCL process and any dependent processes.

Entering NonStop
NET/MASTER MS

Commands From an
NCL Process

If you want to enter a NonStop NET/MASTER MS command from an NCL process,
you must use either the CMD core statement or the INTCMD verb.  The difference
between CMD and INTCMD lies in the environment to which each sends the NonStop
NET/MASTER MS command for execution.  This in turn affects where the results of
the NonStop NET/MASTER MS command are returned, which determines whether
the NCL process that executes CMD or INTCMD receives the results.

This subsection explains how to enter NonStop NET/MASTER MS commands,
including the START and EXEC commands, from an NCL process, using the CMD
core statement and the INTCMD verb.  It discusses command entry restrictions from
an NCL process.

Entering Commands Using
the CMD Core Statement

The CMD core statement sends the NonStop NET/MASTER MS command that
follows it for execution in the same environment in which the NCL process is
executing.  From the point of view of the NCL process, this is the current execution
environment.  The NonStop NET/MASTER MS command and the NCL process are
processed asynchronously.

The results are not returned to the NCL process that executes CMD:  this NCL process
never receives the results.  Where the results are returned, depends on the type of
current execution environment in which the NCL process is executing:  for example, if
the NCL process is executing in a primary processing environment attached to an OCS
window, the results are sent to the OCS window.

The general syntax of the CMD core statement is:

CMD nnm-command

nnm-command

specifies the name of a NonStop NET/MASTER MS command.

Figure 16-7 shows how the CMD core statement works in the current execution
environment.

Figure 16-7.  Using CMD in the Current Execution Environment

START/EXEC

Current Execution Environment

•  Remote System

•  OCS Window

•  Monitor  Terminal

•  Activity Log

•  NCL Process

Results

NCL Process

017

Can Go to

CMD NNM Command Processing

Note Figure 16-7 does not show the OCS command input line prompt (=>) because the current execution
environment is not necessarily the OCS primary processing environment.

It shows that CMD, conceptually, executes NonStop NET/MASTER MS commands on
the same level as the NCL process that executes CMD.

Figure 16-8 shows how the CMD core statement works when it executes in the primary
processing environment associated with an OCS window.

Figure 16-8.  Using CMD in the Primary Processing Environment Under OCS

Note Figure 16-8 shows the OCS command input line prompt (=>) because the primary processing
environment is the OCS primary processing environment.

Again, this shows that CMD, conceptually, executes NonStop NET/MASTER MS
commands on the same level as the NCL process that executes CMD.

The following NCL procedure uses the CMD core statement to execute a NonStop
NET/MASTER MS command:

zex1601n: PROCEDURE
   /* Uses CMD to execute a command.  The */
   /* command is entered as a parameter.  */
   &command = &SYS.ALLPARMS
   SAY The command you have entered is &command
   CMD &command
END zex1601n

The command is entered as a parameter when the NCL procedure is executed and is
placed in the system variable &SYS.ALLPARMS.

The following screen shows the results of executing the procedure, passing the SHOW
OCS command to it as a parameter:

  (11:03) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1601N SHOW OCS
   The command you have entered is SHOW OCS
   SHOW OCS
   NNM0398 -USERID- TERMINAL --------NAME-------- ------LOCATION------ STATUS
   NNM0399 NMTJN    #3362826 JOHN NEW             PUBLICATIONS         ACTIVE
   NNM0999 *END*
   NNM1005 START ZEX1601N PROCESSING COMPLETE. NCLID 000032
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Since the SHOW OCS command is executed in the current execution environment, the
NCL process does not receive the results of the command.  Since the current execution
environment is the primary processing environment associated with the OCS window,
the results are sent to the OCS window for display.  Using this NCL procedure to
execute the SHOW OCS command has the same result as entering the SHOW OCS
command from the OCS command input line.

Entering Commands Using
the INTCMD Verb

The INTCMD verb sends the NonStop NET/MASTER MS command that follows it for
execution in the dependent processing environment associated with the NCL process.
The NonStop NET/MASTER MS command and the NCL process are processed
asynchronously.

The results are then made available to the NCL process that executes INTCMD.  The
NCL process can either read the results, using the INTREAD verb, or discard the
results, using the INTCLEAR verb.  After the NCL process reads the results, it can
send the results to the next higher processing environment by using the INTCONT
verb.

The general syntax of the INTCMD verb is:

INTCMD nnm-command

nnm-command

specifies the name of a NonStop NET/MASTER MS command.

Figure 16-9 shows how INTCMD executes commands in the dependent processing
environment associated with an NCL process.  It shows how INTREAD reads each
message of the results and how INTCONT passes each message to the next higher
processing environment.  (In Figure 16-9 the next higher environment is the primary
processing environment associated with an OCS window.)

Figure 16-9.  Using INTCMD to Execute a Command

The following NCL procedure uses the INTCMD verb to execute a NonStop
NET/MASTER MS command:

zex1602n: PROCEDURE
   /* Uses INTCMD to execute a command.  The */
   /* command is entered as a parameter.     */
   &command = &SYS.ALLPARMS
   SAY The command you have entered is &command
   INTCMD &command
END zex1602n

The command is entered as a parameter when the NCL procedure is executed and is
placed in the system variable &SYS.ALLPARMS.

The following screen shows the results of executing the procedure:

  (11:11) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1602N SHOW OCS
   The command you have entered is SHOW OCS
   NNM1005 START ZEX1602N PROCESSING COMPLETE. NCLID 000035
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Since the SHOW OCS command is executed in the dependent processing environment
associated with the NCL process, the results are available to the NCL process.
However, since the NCL process terminates execution before reading the results with
INTREAD, the results are not sent to the OCS window for display.

The following NCL procedure uses the INTREAD verb to read each line of the results
of a command, and the INTCONT verb to send each line of the results to the next
higher processing environment:

zex1603n: PROCEDURE
   /* Uses INTCMD to execute a command.  The */
   /* command is entered as a parameter.     */
   &command = &SYS.ALLPARMS
   SAY The command you have entered is &command
   INTCMD &command
   DO UNTIL &msgno = NNM0999
      INTREAD VARS=&msgno
      INTCONT
   END
END zex1603n

The following screen shows the results of executing the procedure:

  (14:17) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1603N SHOW OCS
   The command you have entered is SHOW OCS
   NNM0398 -USERID- TERMINAL --------NAME-------- ------LOCATION------ STATUS
   NNM0399 NMTJN    #5156654 John New             PUBLICATIONS         ACTIVE
   NNM0999 *END*
   NNM1005 START ZEX1603N PROCESSING COMPLETE. NCLID 000089
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

The NCL process uses a DO loop and the INTREAD verb to read each line of the
results of the SHOW OCS command.  It uses the INTCONT verb to send each line to
the primary processing window associated with the OCS window from which the
NCL procedure was executed.  The DO loop (and the NCL process) terminates when
the loop detects an end message with a message number of NNM0999.  Using this
NCL procedure to execute the SHOW OCS command has the same result as entering
the SHOW OCS command from the OCS command input line.

Executing START and
EXEC Using CMD and

INTCMD

START and EXEC are NonStop NET/MASTER MS commands that execute an NCL
procedure.  Any NCL process can execute combinations of CMD and INTCMD, with
START and EXEC:

CMD START
CMD EXEC
INTCMD START
INTCMD EXEC

Figure 16-5, earlier in this section, shows the result of an NCL process executing CMD
START and CMD EXEC in a primary processing environment.  The results of every
NCL process executing in this current execution environment flow to the OCS
window.

Figure 16-10 shows the result of an NCL process executing INTCMD START and
INTCMD EXEC.  It assumes that the top-level environment is a primary processing
environment.  It shows that the results of every NCL process flow to its owner.  In
particular, notice how the output from a command can flow to an OCS window.

Figure 16-10.  Executing INTCMD START and INTCMD EXEC

The following three NCL procedures show the flow of messages depicted in
Figure 16-10.  They illustrate how INTCMD passes a command to a dependent
processing environment, and how INTREAD and INTCONT pass messages up the
environmental hierarchy.

zex1604n: PROCEDURE
   /* Uses INTCMD to execute START */
   SAY Primary processing environment
   INTCMD START ZEX1605N
   DO UNTIL &msgno = NNM0999
      INTREAD VARS=&msgno
      INTCONT
   END
   SAY Terminating ZEX1604N
END zex1604n

zex1605n: PROCEDURE
   /* Uses INTCMD to execute EXEC */
   SAY First dependent processing environment
   INTCMD EXEC ZEX1606N
   DO UNTIL &msgno = NNM0999
      INTREAD VARS=&msgno
      INTCONT
   END
END zex1605n

zex1606n: PROCEDURE
   /* Uses INTCMD to execute SHOW OCS */
   SAY Second dependent processing environment
   INTCMD SHOW OCS
   DO UNTIL &msgno = NNM0999
      INTREAD VARS=&msgno
      INTCONT
   END
END zex1606n

The following screen shows the result of executing the first NCL procedure.  The first
NCL procedure then executes the second NCL procedure using INTCMD START.  The
second NCL procedure then executes the third NCL procedure using INTCMD EXEC.
The third NCL procedure then executes the SHOW OCS command using INTCMD
SHOW OCS.

  (14:19) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1604N
   Primary processing environment
   First dependent processing environment
   Second dependent processing environment
   NNM0398 -USERID- TERMINAL --------NAME-------- ------LOCATION------ STATUS
   NNM0399 NMTJN    #5156654 John New             PUBLICATIONS         ACTIVE
   NNM0999 *END*
   Terminating ZEX1604N
   NNM1005 START ZEX1604N PROCESSING COMPLETE. NCLID 000090
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

The results of the SHOW OCS command are passed from NCL process to NCL process
using the INTREAD and INTCONT verbs.  Finally, the results are passed to the OCS
window for display.  Using these NCL procedures to execute the SHOW OCS
command and to process the results has the same result as entering the SHOW OCS
command from the OCS command input line.

Combinations of CMD and
INTCMD With START and

EXEC

Figure 16-11 combines the concepts of Figure 16-9 and Figure 16-10.  The figure
assumes that the highest environment is a primary processing environment.  It shows
that the results of the CMD core statement always flow to the owner of the next higher
level in the environmental hierarchy.  At the highest level, this is the OCS window;  at
lower levels, an NCL process.  The figure shows that the results of the INTCMD verb
always flow to the NCL process that owns the dependent processing environment.

Figure 16-11.  Combinations of CMD, INTCMD, START, and EXEC

 ==> START/EXEC

Primary Processing Environment

OCS Window (Message Area)

Results

NCL
CMD NNM

CMD EXEC/
         START

NCL
CMD NNM

CMD EXEC/
         START

NCL
CMD NNM

CMD EXEC/
         START

ResultsResults

INTCMD
START/EXEC

Dependent Processing Environment

Results

NCL
CMD NNM

CMD EXEC/
         START

NCL
CMD NNM

CMD EXEC/
         START

NCL
CMD NNM

CMD EXEC/
         START

ResultsResults

INTCMD
START/EXEC

Dep. Proc. Env.

NCL
CMD NNM

CMD EXEC/
         START

Results Results

NCL
CMD NNM

CMD EXEC/
         START

NCL
CMD NNM

CMD EXEC/
         START

Results

INTCMD
START/EXEC

INTCMD
START/EXEC

• • •

Dep. Proc. Env.

INTCMD
START/EXEC

INTCMD
START/EXEC

Dep. Proc. Env.

NCL

Dep. Proc. Env.

NCL

Dep. Proc. Env.

NCL
021



Entering NonStop NET/MASTER MS Commands From an NCL Process

Environments and Command Processing

106160 Tandem Computers Incorporated 16–29

NonStop NET/MASTER MS
Command Entry

Restrictions

When you enter a NonStop NET/MASTER MS command from an NCL procedure,
you must enclose the command in quotes if it contains operators or special characters.
The equal sign (=) is an example of a character that NCL uses as an operator.  Refer to
the NonStop NET/MASTER NCL Reference Manual for a complete list of operators and
special characters.

You can enclose in quotes either the complete command string or just those parts that
contain the operator(s) or special character(s).  The following examples show
commands that must be enclosed in quotes when entered from an NCL procedure
because they contain an equal sign:

LINK DEFINE=ZNNM MSGID=ZNNM COLOR=BLUE
SESSION DEFINE=ZNNMS LINK=ZNNM TYPE=PTP
UNIT DEFINE=ZNNMU SESSION=ZNNMS MODE=PRIMARY,
     DEVICE=$ZNNM.SUBVOL1.MAILBOX1

The following examples show how to correctly enter these commands from an NCL
procedure:

CMD "LINK DEFINE=ZNNM MSGID=ZNNM COLOR=BLUE"
CMD "SESSION DEFINE=ZNNMS LINK=ZNNM TYPE=PTP"
CMD "UNIT DEFINE=ZNNMU SESSION=ZNNMS MODE=PRIMARY",
       "DEVICE=$ZNNM.SUBVOL1.MAILBOX1"

Note The presence of the dollar sign ($), period (.), and number sign (#) in these examples does not mean that
the command must be enclosed in quotes;  these characters are not operator characters or special
characters.

You should be particularly careful if you include a logical name as part of a NonStop
NET/MASTER MS command within an NCL procedure.  Logical names can contain
any displayable character, including characters that NCL regards as operators and
special characters.  For example, if you want to refer to a terminal that has the logical
name of TERM(1), you must enclose it in quotes because the parentheses are special
characters:

CMD "TERMINAL ADD=TERM(1) DEVICE=$ATP2.#TERM1"

Note As a general rule, always enclose NonStop NET/MASTER MS commands in quotes when you enter them
from an NCL procedure.

You can execute most NonStop NET/MASTER MS commands from an NCL
procedure by using either the CMD core statement or the INTCMD verb.  Some
commands cannot be executed from an NCL procedure by using the CMD core
statement;  others cannot be executed by using the INTCMD verb;  and a few cannot
be executed from an NCL procedure at all.  Refer to the NonStop NET/MASTER MS
Command Reference Manual to determine CMD and INTCMD command entry
restrictions for individual commands.

Overriding Command
Display and Logging

When you enter a command from an NCL process by using the CMD core statement
or the INTCMD verb, you can control command echoing and logging.

You can stop a specific command from being echoed to the roll-delete message display
area and from being logged in the NonStop NET/MASTER MS activity log by
preceding the command with the suppression character (-).  By using the suppression
character, you display and log only the results of the command.  The following
example executes a PROFILE command, but prevents the command from being
displayed and logged:

CMD "-PROFILE CMDKEEP=NO NCLTEST=YES"

The following example prevents both the K and SHOW USERS commands from being
displayed and logged:

CMD "-K;-SHOW USERS"

You can stop a sequence of commands from being echoed to the roll-delete message
display area, but not from being logged in the activity log, by preceding the sequence
with the CONTROL NOCMD verb.  By using this verb, you prevent echoing of a
sequence of commands from an NCL process until the NCL process terminates or
executes the CONTROL CMD verb.

The Dependent
Request and

Response Queues

The preceding discussion of dependent processing environments did not discuss one
important subject—the request and response queues that are associated with the
dependent processing environment of every NCL process.

Accordingly, this subsection builds upon the earlier discussion of dependent
processing environments.  It explains the purpose of the request and response queues.
In particular, it discusses how to use the INTREAD, INTCLEAR, and INTCONT verbs
in more detail.

The Purpose of the
Request and Response

Queues

Every NCL process can have a dependent processing environment associated with it.
Since every dependent processing environment has two queues associated with it,
every NCL process can have two queues associated with it.  These queues are called
the request queue and the response queue.

These queues are automatically created and deleted whenever an NCL process—and
the dependent processing environment associated with it—is created and deleted.

The purpose of the request and response queues is to hold event messages.  The
following table summarizes the type of message that each queue holds by default:

Dependent Queue Message Type

Request Messages sent from an OCS operator using the INTQ command.

Response Messages sent from an NCL process using the WRITE verb; and results of
commands executed by an NCL process using the INTCMD verb.

Using two queues to hold different types of event messages makes it easier for you to
process each type of message.

Figure 16-12 shows that every dependent processing environment has a request and
response queue.  It also shows that the results of a NonStop NET/MASTER MS
command are sent by default to the response queue.  Figure 16-12 supplements the
information presented earlier in this section in Figure 16-10.

Figure 16-12.  The Request and Response Queues

The INTREAD, INTCONT,
and INTCLEAR Verbs

Three verbs are available to work with messages that arrive in the request and
response queues:

INTREAD
INTCLEAR
INTCONT

Figure 16-13 shows the use of these verbs in a primary processing environment
(although it could be any type of current execution environment).

Figure 16-13.  Using INTCMD, INTREAD, INTCLEAR, and INTCONT

Reading a Message From a Dependent Queue

You can use the INTREAD verb to read a message from either the request or response
queue.

When the INTREAD verb reads a message it automatically places the text and the
attributes of the message into an MDO variable beginning with the stem &$INT..  The
names of all the structures and substructures contained in the MDO variable begin
with this stem.  See Section 10, “Working With Mapping Services,” for a detailed
discussion of MDO variables.

When the INTREAD verb reads a message, it has the option of placing the text (but not
the attributes of the text) into ordinary variables, using keywords such as ARGS,
VARS, RANGE, SEGMENT, and PARSE.  Refer to the discussion on verb syntax and
variable access methods in the NonStop NET/MASTER NCL Reference Manual for more
information about these keywords.

You can specify the type of queue from which the INTREAD verb reads a message by
using the TYPE keyword.  The following table summarizes the meaning of the
qualifiers that can follow the TYPE keyword:

TYPE Qualifier Description

TYPE=ANY Allows the INTREAD verb to read messages from either the request or response
queue.

TYPE=REQ Allows the INTREAD verb to read messages from only the request queue.

TYPE=RESP Default.  Allows the INTREAD verb to read messages from only the response
queue.

Note The system variable &SYS.LOOPCTL is set to 1000 each time the INTREAD verb is executed (unless it is
explicitly set to 0 or to a number greater than 1000 by the NCL process).

The following two NCL procedures show how INTCMD passes a command to a
dependent processing environment and how INTREAD and INTCONT pass messages
up the environmental hierarchy.  The INTREAD verb in each NCL procedure
explicitly specifies the response queue.

zex1607n: PROCEDURE
   /* Procedure to analyze INTREAD messages */
   INTCMD START ZEX1608N
   DO UNTIL &msgno1 = NNM0999
      INTREAD VARS=&msgno* TYPE=RESP
      IF &msgno1 = "NNM0999" THEN DO
         SAY "&$INT.TEXT is "&$int.text
         SAY "&$INT.MSGATTR.ALARM is "&$int.msgattr.alarm
         SAY "&$INT.MSGATTR.DISPLAY.COLOR is "&$int.msgattr.display.color
         SAY "&$INT.MSGATTR.DISPLAY.BKGCOLOR is "&$int.msgattr.display.bkgcolor
         SAY "&$INT.MSGATTR.DISPLAY.HLITE is "&$int.msgattr.display.hlite
         SAY "&$INT.MSGATTR.DISPLAY.INTENS is "&$int.msgattr.display.intens
         SAY "&$INT.SOURCE.TIME is "&$int.source.time
         SAY "&$INT.SOURCE.USER is "&$int.source.user
         DO &i = 1 TO &SYS.VARCNT
            SAY "Word "&i" of text is "&msgno&i
            END
         INTCONT
         END /*do*/
      ELSE
         INTCONT
   END /*do*/
END zex1607n

zex1608n: PROCEDURE
   /* Uses INTCMD to execute SHOW OCS */
   INTCMD SHOW OCS
   DO UNTIL &msgno = NNM0999
      INTREAD VARS=&msgno TYPE=RESP
      INTCONT
   END
END zex1608n

The following screen shows the result of executing the first NCL procedure.  The first
NCL procedure then executes the second NCL procedure using INTCMD START.  The
second NCL procedure then executes a command, SHOW OCS, using INTCMD
SHOW OCS.

  (12:03) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1607N
   NNM0398 -USERID- TERMINAL --------NAME-------- ------LOCATION------ STATUS
   NNM0399 NMTMG    #5648197 Mike                 PUBLICATIONS         ACTIVE
   NNM0399 NMTMG    #5648197 Mike                 PUBLICATIONS         ACTIVE
   NNM0399 NMTJN    #5648202 John New             PUBLICATIONS         ACTIVE
   NNM0399 NMTKF    #5648196 KEN FINLAYSON                             ACTIVE
   &$INT.TEXT is NNM0999 *END*
   &$INT.MSGATTR.ALARM is 0
   &$INT.MSGATTR.DISPLAY.COLOR is NONE
   &$INT.MSGATTR.DISPLAY.BKGCOLOR is NONE
   &$INT.MSGATTR.DISPLAY.HLITE is NONE
   &$INT.MSGATTR.DISPLAY.INTENS is NORMAL
   &$INT.SOURCE.TIME is 1994/05/20 12:03:26.115
   &$INT.SOURCE.USER is NMTMG
   Word 1 of text is NNM0999
   Word 2 of text is *END*
   NNM0999 *END*
   NNM1005 START ZEX1607N PROCESSING COMPLETE. NCLID 000564
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

The results of the SHOW OCS command are passed from NCL process to NCL process
using the INTREAD and INTCONT verbs.  Finally, the results are passed to the OCS
window for display.  The INTREAD verb uses the VARS keyword to extract the first
word of each message on the response queue, typically the message number.  If the
first word is the end message that ends many groups of NonStop NET/MASTER MS
messages—NNM0999 *END*—it displays the contents of some of the substructures in
the MDO variable—&$INT.—that holds the message.

Passing on a Message From a Dependent Queue

After reading a message using the INTREAD verb, you can use the INTCONT verb to
pass the message to the next-higher-level environment.

You can use the INTCONT verb to change some attributes of the message before the
verb passes the message on.  Using the INTCONT verb to change attributes updates
fields in the MDO variable, &$INT., that contains the message.  The following table
summarizes the operands of the INTCONT verb that change attributes and the fields
that are updated in &$INT.:

INTCONT Operand &$INT. Field Updated

COLOR &$INT.MSGATTR.DISPLAY.COLOR

BKGCOLOR &$INT.MSGATTR.DISPLAY.BKGCOLOR

HLITE &$INT.MSGATTR.DISPLAY.HLITE

INTENS &$INT.MSGATTR.DISPLAY.INTENS

ALARM &$INT.MSGATTR.ALARM

NRD &$INT.NRD.TYPE
&$INT.DOMID

DATA &$INT.TEXT

Note If an INTCONT verb does not change message attributes, you cannot improve the performance of an
NCL process by omitting it.  The next INTREAD verb does not perform an implicit INTCONT.  You must
explicitly use INTCONT to pass on a message to the next higher-level environment; otherwise, the
message is lost.

The following two NCL procedures show the flow of messages to and from a
dependent processing environment.  They illustrate how INTCMD passes a command
to a dependent processing environment and how INTREAD and INTCONT pass
message up the environmental hierarchy.  The INTREAD verb in each NCL procedure
explicitly specifies the response queue.

zex1609n: PROCEDURE
   /* Procedure to analyze INTREAD messages */
   INTCMD START ZEX1610N
   DELAY 10
   DO &i = 1 TO &SYS.INT.RSP#
      INTREAD VARS=&msgno* TYPE=RESP
      IF &msgno1 = "NNM0398" THEN DO
         SAY "&$INT.MSGATTR.DISPLAY.COLOR is "&$int.msgattr.display.color
         SAY "&$INT.MSGATTR.DISPLAY.BKGCOLOR is "&$int.msgattr.display.bkgcolor
         SAY "&$INT.MSGATTR.DISPLAY.HLITE is "&$int.msgattr.display.hlite
         INTCONT
      END /*do*/
      ELSE
         INTCONT
   END /*do*/
END zex1609n

zex1610n: PROCEDURE
   /* Uses INTCMD to execute SHOW OCS */
   INTCMD SHOW OCS
   DELAY 5
   DO &i = 1 TO &SYS.INT.RSP#
      INTREAD VARS=&msgno TYPE=RESP
      IF &msgno = "NNM0398" THEN
         INTCONT COLOR=BLUE HLITE=BLINK
      ELSE
         INTCONT
   END /*do*/
END zex1610n

The following screen shows the result of executing the first NCL procedure.  The first
NCL procedure then executes the second NCL procedure using INTCMD START.  The
second NCL procedure then executes a command, SHOW OCS, using INTCMD
SHOW OCS.

  (12:13) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START zex1609n
   &$INT.MSGATTR.DISPLAY.COLOR is BLUE
   &$INT.MSGATTR.DISPLAY.BKGCOLOR is NONE
   &$INT.MSGATTR.DISPLAY.HLITE is BLINK
   NNM0398 -USERID- TERMINAL --------NAME-------- ------LOCATION------ STATUS
   NNM0399 NMTMG    #5648197 Mike                 PUBLICATIONS         ACTIVE
   NNM0399 NMTMG    #5648197 Mike                 PUBLICATIONS         ACTIVE
   NNM0399 NMTJN    #5648202 John New             PUBLICATIONS         ACTIVE
   NNM0399 NMTKF    #5648196 KEN FINLAYSON                             ACTIVE
   NNM0999 *END*
   NNM1005 START ZEX1610N PROCESSING COMPLETE. NCLID 000569
   NNM1005 START ZEX1609N PROCESSING COMPLETE. NCLID 000568
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
    =>

The results of the SHOW OCS command are passed from NCL process to NCL process
using the INTREAD and INTCONT verbs.  The INTCONT verb in ZEX1610N modifies
attributes of message NNM0398.  The ZEX1609N NCL procedure displays the value of
the fields in &$INT. that are changed.

When a group of messages arrives in a request or response queue, the system variable
&SYS.INT.RSP# or &SYS.INT.REQ#, respectively, contains the number of messages.
The NCL procedures use the variable &SYS.INT.RSP# as the terminating value in the
DO loops that read messages.  ZEX1610N uses the DELAY verb to ensure that all
messages in the group arrive in the queue before &SYS.INT.RSP# is updated;  the
delay ensures the correct number of iterations of the DO loop.

The &SYS.INTYPE system variable contains the name of the dependent queue on
which a message has arrived.  &SYS.INTYPE contains REQ if the message arrived on
the request queue, or RESP if the message arrived on the response queue.  If there is no
message, &SYS.INTYPE contains a null value.

Note Command processing and NCL process execution is asynchronous.  Since messages from certain
commands such as the SHOW PARAM, SHOW SYSPARMS, and STATUS command may take some
time to arrive, you must ensure that message arrival and message reading is synchronized.  In addition to
the DELAY verb, you can use the WAIT operand in the INTREAD verb to control the timing of message
reading.  Use of the DELAY verb and the WAIT operand of the INTREAD verb can affect the performance
of an NCL process.

Clearing Messages in Dependent Queues

You can use the INTCLEAR verb to clear messages in the request and response
queues.  You can specify the type of queue using the TYPE keyword.  The following
table summarizes the meaning of the qualifiers that can follow the TYPE keyword:

TYPE Qualifier Description

TYPE=ALL This is the default.  Clears all messages in the request and response queues.
Dismantles the dependent processing environment.  Terminates all commands
and NCL processes running in the environment.

TYPE=ANY Clears all messages in the request and response queues.  Does not dismantle the
dependent processing environment.  Does not terminate commands and NCL
processes running in the environment.

TYPE=REQ Clears all messages in the request queue.  Does not dismantle the dependent
processing environment.  Does not terminate commands and NCL processes
running in the environment.

TYPE=RESP Clears all messages in the response queue.  Does not dismantle the dependent
processing environment.  Does not terminate commands and NCL processes
running in the environment.

The following NCL procedure executes the STATUS command in a dependent
processing environment.  It reads the first five messages that arrive on the response
queue and then clears all other messages on the queue.

zex1611n: PROCEDURE
   /* Procedure to analyze INTREAD messages */
   INTCMD STATUS
   DO 5
      INTREAD
      INTCONT
   END
   INTCLEAR
END zex1611n

The following screen shows the results of executing the procedure:

  (09:26) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1611N
   STATUS
   NNM0608 SYSTEM STATUS AS AT 09:26:32
   NNM0608 GUARDIAN VERSION: D30
   NNM0608 SYSTEM STATE: RUNNING
   NNM0608 SYSTEM WAS INITIALISED ON MON 15-JUN-1992 AT 09:22:55
   NNM1005 START ZEX1611N PROCESSING COMPLETE. NCLID 000007
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Example With the INTCMD, INTREAD, and INTCLEAR Verbs

The following NCL procedure shows a more complex example of the INTCMD,
INTREAD, and INTCLEAR verbs.  (The INTCONT verb is not required;  the SAY core
statement is used to display results.)

zex1612n: PROCEDURE
   /* Get single NonStop NET/MASTER PARAM value using the INT* verbs   */
   /* Change NCL procedure name from ZEX1612N to PARAM to enter the    */
   /* "PARAM" command from the OCS command input line.                 */
   CONTROL SHRVARS=&param                      /* Set variable sharing */
   &param = &SYS.ALLPARMS                      /* Initialize variable  */
   IF &param = "" THEN                         /* If no PARAM specified*/
      CMD "-SHOW PARAM"                        /* display all PARAMs   */
   ELSE                                        /* else                 */
      SAY &param" is "get_value(&param)        /* display value        */
   get_value: FUNCTION                         /* Function to get value*/
      CONTROL NOCMD                            /* Disable command echo */
      INTCLEAR TYPE=ALL                        /* Clear dep. queues    */
      INTCMD "SHOW PARAM"                      /* Execute SHOW PARAM   */
      DO UNTIL &msgno = "NNM0999" OR,          /* DO loop sets exit    */
            &found > 0                         /* conditions           */
         INTREAD VARS=(&msgno(7),*,&text),     /* Read message into    */
            TYPE=ANY PARSE=NO                  /* variables            */
         &found = POS(&param"=",&text)         /* Check for PARAM      */
         IF &found > 0 THEN                    /* If PARAM found then  */
            &value = SPACE(REMSTR('=',&text))  /* get value            */
         ELSE                                  /* else                 */
            &value = "not a valid value"       /* not valid            */
      END                                      /* End DO loop          */

      RETURN(&value)                           /* Return proc char     */
   END get_value                               /* End function         */
END zex1612n

This NCL procedure allows you to obtain the value of a single startup parameter.
(Although you can display the value of single system parameters using the SHOW
SYSPARMS command, you cannot display the value of a single startup parameter
using the SHOW PARAM command.)  Comments throughout the NCL procedure
describe each step.

The following screen shows the result of executing the NCL procedure five times:

  (12:38) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1612N XYZ
   XYZ is not a valid value
   NNM1005 START ZEX1612N PROCESSING COMPLETE. NCLID 000067
   START ZEX1612N CIP
   CIP is not a valid value
   NNM1005 START ZEX1612N PROCESSING COMPLETE. NCLID 000068
   START ZEX1612N CIP[1]
   CIP[1] is $ZCIP (EXTERNAL)
   NNM1005 START ZEX1612N PROCESSING COMPLETE. NCLID 000069
   START ZEX1612N CONFIG FILE
   CONFIG FILE is $DATA9.ZNNM.CONFIG
   NNM1005 START ZEX1612N PROCESSING COMPLETE. NCLID 000070
   START ZEX1612N
   NNM0605 CONFIG FILE=$DATA9.ZNNM.CONFIG
   NNM0605 PROCESSCHAR=Z
   NNM0605 INIT=INIT
   NNM0605 READY=ZREADY
   NNM0605 SWGUID=YES
   NNM0605 DID=SYS1
   NNM0605 NCLDISTSRC=$DATA9.ZNNMNDS
   NNM0605 NCLCUSTSRC=$DATA9.ZNNMNCS
 MSG QUEUED ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Sending a Message to
an NCL Process, using

the INTQ Command

An OCS operator can use the NonStop NET/MASTER MS INTQ command from the
OCS command input line to send a message to the dependent request or response
queue of a target NCL process.

Note You should use the WRITE verb, rather than CMD INTQ or INTCMD INTQ, to send a message from
another NCL process to another NCL process.  This use of the WRITE verb is discussed later in this
section.

You must specify the NCL ID of the target NCL process in the INTQ command if there
is more than one NCL process executing in the current execution environment or you
are sending the message to an NCL process executing in an environment you do not
own.

The message sent by the INTQ command is treated as a series of parameters, delimited
by blanks.  Each parameter can be placed in a separate variable when the message is
read by the INTREAD verb.

Note The INTQ command sends constant data to an NCL process.  You should not use the INTQ command to
send a variable, such as an MDO variable, to an NCL process;  otherwise, unpredictable results may
occur.  You should use the WRITE verb to do this.  This use of the WRITE verb is discussed later in this
section.

By default, the INTQ command sends data to the request queue of an NCL process
(TYPE=REQ).  You can specify TYPE=RESP to send data to a response queue.

By default, the INTREAD verb reads messages only from the response queue.  You
must specify INTREAD TYPE=REQ if you want the INTREAD verb to read a message
only from the request queue.  You can specify INTREAD TYPE=ANY if the INTREAD
verb can read messages from either dependent queue.

Note It is your responsibility to ensure that the INTQ command sends messages to the correct queue and that
the INTREAD verb reads messages from the correct queue.  You must synchronize INTQ and INTREAD.

You can use the INTQ command to send a message to any NCL process that you own:
that is, any NCL process that is owned by your user ID.  If you have sufficient
authority you can use INTQ to send a message to any NCL process in any region, if
the value of the SYSPARMS NCLXUSER system parameter is YES.

Note If you are logged on twice or more with the same user ID, you can use the INTQ command to send a
message from one of your regions to another, regardless of the setting of SYSPARMS NCLXUSER.

Figure 16-14 shows how the INTQ command is used from the OCS command input
line to send a message to an NCL process.  The NCL process must issue INTREAD to
read the message.  (This figure assumes that SYSPARMS NCLXUSER=YES, since a
message is sent from one region to another region that has a different user ID.)

Figure 16-14.  Using INTQ to Send a Message

Region: NNMAEB

OCS Window

OCS Window
==> INTQ ID=555 TYPE=RESP DATA=ABCDEF

Primary Processing Environment

Region: NNMCW

024

OCS Window

OCS Window
==> INTQ ID=555 DATA=123456

Primary Processing Environment

REQ RESP

NCL Process  ID=555

Dep. P.E.

Example of the INTQ
Command

The following NCL procedure shows how to use INTQ from the OCS command input
line to send a message to an NCL process.  This NCL procedure waits for messages to
arrive on its request queue.  It reads each message, using INTREAD.  It passes each
message to the next higher level environment, using INTCONT.  To see how this NCL
procedure works, follow these steps:

1. Type the NCL procedure as shown, calling it ZEX1613N:

zex1613n: PROCEDURE
   /* Receives messages on dep. request queue */
   /* from an INTQ command sent from the OCS  */
   /* command input line.                     */
   ON ERROR FLUSH
   SAY "NCL ID of ZEX1613N is "&SYS.NCLID
   DO FOREVER
      INTREAD WAIT=YES TYPE=REQ
      INTCONT COLOR=RED HLITE=BLINK
   END
END zex1613n

2. Execute ZEX1613N from the OCS command input line using the START
command:

START ZEX1613N

Observe the NCL ID of ZEX1613N, because you must enter this as a parameter
when you use the INTQ command to send it a message.  You will use the NCL ID
to flush the NCL process later.  (You can also determine the NCL ID of an NCL
process by using the SHOW NCL command.)

3. Send a message to ZEX1613N, using the following command from the OCS
command input line:

INTQ ID=nclid DATA=This is a test message from OCS

Replace nclid with the NCL ID you obtained in Step 2.

4. Observe the results of using the INTQ command from the OCS command input
line to send a message to ZEX1613N:

  (12:50) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1613N
   NCL ID of ZEX1613N is 75
   INTQ ID=75 DATA=This is a test message from OCS
   This is a test message from OCS
   FLUSH ID=75
   NNM1022 FLUSH COMMAND ACCEPTED
   NNM1005 START ZEX1613N PROCESSING COMPLETE. NCLID 000075
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=> INTQ ID=75 DATA=This is a test message from OCS

Notice how the INTCONT verb changes the attributes of the message.

Sending a Message
Between NCL

Processes, Using the
WRITE Verb

You can use the WRITE verb, by specifying an NCL ID, to send a message from one
NCL process to the dependent request or response queue of a target NCL process.
When you are sending messages between NCL processes, you should carefully
consider how to send the message and how to read the message.  You can use the
WRITE verb to send:

Text, specified by using the DATA operand.

One or more ordinary variables, specified by using the ARGS or VARS operand
(with or without additional text).

An enclosed MDO variable, specified by using the MDO operand (with or without
additional text).

You can use an INTREAD verb in the target NCL process to read the message from the
dependent request or response queue.  When reading a message into variables, you
can use the INTREAD verb:

Without any variable specification, to read the whole message into the &$INT.
MDO variable.

With either the ARGS or VARS operands, to read either the text or the values in
the ordinary variables into ordinary variables.

With the MDO operand, to read an enclosed MDO variable (if one is present in the
message) into a user-specified enclosing MDO variable.

You can also specify how to parse a message by specifying the PARSE operand in the
INTREAD verb (the default is PARSE=YES).

The following discussion describes the options that are available to you to send a
message from one NCL process to a target NCL process.

You can use the LOGREAD and MSGREAD verbs to read a message into variables in
the same way as the INTREAD verb.  LOGREAD is used in the LOGPROC NCL
procedure:  you can send a message to LOGPROC (if it is active) by specifying WRITE
LOG=YES.  MSGREAD is used in a MSGPROC NCL procedure:  you can send a
message to a MSGPROC NCL procedure (if one is active for an OCS window) by using
a variety of WRITE verb operands.  These verbs are discussed in detail in Section 17,
“Developing System-Level NCL Procedures.”

Sending Text You can send text between NCL processes by using the DATA operand of the WRITE
verb.  The DATA operand is followed by an expression, which can contain either
constant or variable values, or both.

When an INTREAD verb reads a message, the specified text is placed in the
&$INT.TEXT MDO variable.  If the DATA operand is not used or specifies a null
value, the value of &$INT.TEXT is null.  The &$INT.USERMDO and
&$INT.MAPNAME MDO variables are both null.

The following WRITE verb specifies the DATA operand to send text:

WRITE NCLID=&nclid DATA=a b c

The following WRITE verb does not specify the DATA operand and sends no text:

WRITE NCLID=&nclid

The result of reading a message depends on the syntax of the INTREAD verb that you
use;  in particular, whether you specify PARSE=YES or PARSE=NO.

The following table shows the result of using the WRITE verb to send text (with and
without the DATA operand), and the result of using an INTREAD verb that specifies
PARSE=YES to read the message:

WRITE Verb

INTREAD Verb
WRITE NCLID=&nclid
      DATA=a b c

WRITE NCLID=&nclid

INTREAD
PARSE=YES

&$INT.TEXT=a b c
&$INT.USERMDO=null
&$INT.MAPNAME=null

&$INT.TEXT=null
&$INT.USERMDO=null
&$INT.MAPNAME=null

INTREAD
PARSE=YES
ARGS

&$INT.TEXT=a b c
&1=a
&2=b
&3=c
&$INT.USERMDO=null
&$INT.MAPNAME=null

&$INT.TEXT=null
&1=null
&2=null
&3=null
&$INT.USERMDO=null
&$INT.MAPNAME=null

INTREAD
PARSE=YES
MDO=&mdo.

&$INT.TEXT=a b c
&$INT.USERMDO=null
&$INT.MAPNAME=null
&mdo.=null

&$INT.TEXT=null
&$INT.USERMDO=null
&$INT.MAPNAME=null
&mdo.=null

The following table shows the result of using the WRITE verb to send text (with and
without the DATA operand), and the result of using an INTREAD verb that specifies
PARSE=NO to read the message:

WRITE Verb

INTREAD Verb
WRITE NCLID=&nclid
      DATA=a b c

WRITE NCLID=&nclid

INTREAD
PARSE=NO

&$INT.TEXT=a b c
&$INT.USERMDO=null
&$INT.MAPNAME=null

&$INT.TEXT=null
&$INT.USERMDO=null
&$INT.MAPNAME=null

INTREAD
PARSE=NO
ARGS

&$INT.TEXT=a b c
&1=a b c
&2=null
&3=null
&$INT.USERMDO=null
&$INT.MAPNAME=null

&$INT.TEXT=null
&1=null
&2=null
&3=null
&$INT.USERMDO=null
&$INT.MAPNAME=null

INTREAD
PARSE=NO
MDO=&mdo.

&$INT.TEXT=a b c
&$INT.USERMDO=null
&$INT.MAPNAME=null
&mdo.=null

&$INT.TEXT=null
&$INT.USERMDO=null
&$INT.MAPNAME=null
&mdo.=null

Specifying Ordinary Variables When Reading a Message

The following two NCL procedures show how to use the WRITE verb to send text
from one NCL process to a target NCL process that specifies ordinary variables when
reading a message.  To see how these NCL procedures work, follow these steps:

1. Type the first NCL procedure as shown, calling it ZEX1616N:

zex1616n: PROCEDURE
   /* Analyzes messages on the dependent response queue */
   /* sent by the WRITE verb from another NCL process   */
      ON ERROR DO
         &glblzex1616n = ""; FLUSH
      END /*on*/
      /* Uses a global variable to store the NCL ID */
      &glblzex1616n = &SYS.NCLID
      SAY "NCL ID of ZEX1616N is "&SYS.NCLID
      DO FOREVER
         INTREAD ARGS WAIT=YES TYPE=RESP
         DO &cvar = 1 TO &SYS.VARCNT
            SAY "&"&cvar" is "&&cvar
            END /*do*/
         IF &$int.usermdo \= "" THEN DO
            SAY "This message has a USERMDO"
            /* Copy MDO in USERMDO to &MDO. mapped */
            /* by map in MAPNAME                   */
            ASSIGN OPT=VALUE MDO=&mdo. MAP=&$int.mapname,
               DATA=&$int.usermdo
            END /*do*/
         ELSE DO
            SAY "This message does not have a USERMDO"
            /* Copy &$INT. to &MDO. mapped by $MSG */
            ASSIGN OPT=VALUE MDO=&mdo. MAP=$MSG,
               FROM MDO=&$int.
            END /*do*/
         SAY "&$INT.TEXT is "&$int.text
         SAY "&$INT.USERMDO is "&$int.usermdo
         SAY "&$INT.MAPNAME is "&$int.mapname
         SAY "Elements and fields in &MDO. are"
         CALL show_mdo( 0, "" ) SHARE &mdo.
         INTCONT
      END /*do forever*/
      EXIT
   show_mdo: PROCEDURE
      /* Procedure to display all elements and fields */
      /* in an MDO variable regardless of its map     */
      &nspaces = &1
      &piece = &2
      &spaces = COPIES( " ", &nspaces )
      IF &piece \== "" THEN
         &piecedot = &piece"."

      ASSIGN OPT=MDOELEMENTS ARGS FROM MDO=&mdo.&piece
      DO &i = 1 TO &SYS.VARCNT
         &y = &piecedot || &&i
         SAY &spaces || Element &&i is &mdo.&y
         CALL show_mdo( &nspaces + 2, &y ) SHARE &mdo.
         END
      ASSIGN OPT=MDOFIELDS ARGS FROM MDO=&mdo.&piece
      DO &i = 1 TO &SYS.VARCNT
         &y = &piecedot || &&i
         SAY &spaces || Field &&i is &mdo.&y
         CALL show_mdo( &nspaces + 2, &y ) SHARE &mdo.
         END
   END show_mdo
END zex1616n

This procedure is designed to loop forever to trap messages arriving on its
response queue and to analyze their contents.  The INTREAD verb specifies
ordinary variables when reading a message.  The message is analyzed to
determine whether it contains an enclosed MDO variable.  Information about the
message is then displayed on the screen.

2. Type the second NCL procedure as shown, calling it ZEX1617N:

zex1617n: PROCEDURE
   /* Sends text to an NCL process         */
   /* Wait until ZEX1616N begins execution */
   DO UNTIL &glblzex1616n \= ""
      DELAY 1
   END /*do*/
   WRITE COLOR=WHITE,
         NRD=OPER,
         NCLID=&glblzex1616n,
         DATA=a b c
END zex1617n

This procedure sends text to the NCL process whose NCL ID is in the
&GLBLZEX1616N global variable.

Note The use of global variables is not recommended in a production environment since it may impede the
performance of NonStop NET/MASTER MS.

3. Execute ZEX1616N from the OCS command input line, using the following START
command:

START ZEX1616N

When executed, ZEX1616N places its NCL ID in the &GLBLZEX1616N global
variable.

4. Execute ZEX1617N from the OCS command input line, using the following START
command:

START ZEX1617N

5. Observe the results of using the WRITE verb from ZEX1617N to send text to
ZEX1616N, as the following screen shows:

  (13:40) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1616N
   NCL ID of ZEX1616N is 265
   START ZEX1617N
   &1 is a
   &2 is b
   &3 is c
   This message does not have a USERMDO
   &$INT.TEXT is a b c
   &$INT.USERMDO is
   &$INT.MAPNAME is
   Elements and fields in &MDO. are
   Element TEXT is a b c
   Element MSGATTR is ...   ..        .. ".                    .
     Element DISPLAY is .
       Field BKGCOLOR is NONE
       Field BKGCOLOR# is 0
       Field COLOR is WHITE
       Field COLOR# is -9
       Field HLITE is NONE
       Field HLITE# is 0
       Field INTENS is NORMAL
 MSG QUEUED ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

6. Flush ZEX1616N, using the following FLUSH command:

FLUSH ID=nclid-of-zex1616n-ncl-process

Flushing ZEX1616N also deletes the &GLBLZEX1616N global variable.

Specifying an MDO Variable When Reading a Message

The following two NCL procedures show how to use the WRITE verb to send text
from one NCL process to another NCL process (a MSGPROC NCL process in this
example) that specifies an MDO variable when reading a message.  To see how these
NCL procedures work, follow these steps:

1. Type the first NCL procedure as shown, calling it ZEX1620N:

zex1620n: PROCEDURE
   /* MSGPROC to analyze messages containing a USERMDO element, */
   /* which indicates it contains an enclosed MDO variable.     */
   DO FOREVER
      /* If message has USERMDO, put in &mdo. */
      MSGREAD MDO=&mdo.
      IF &$msg.usermdo \= "" THEN DO
         SAY "This message has a USERMDO mapped by "&$msg.mapname
         /* $MSG and $NCL are two most common maps used */
         SELECT &$msg.mapname
            WHEN $MSG THEN DO
               SAY "&MDO.TEXT = "&mdo.text
               /* Parse text into variables */
               PARSE ARGS DATA=&mdo.text
               DO &cvar = 1 TO &SYS.VARCNT

                  SAY "&"&cvar" is "&&cvar
                  END /*do*/
               END /*do*/
            WHEN $NCL THEN DO
               SAY "&MDO. = "&mdo.
               /* How many fields are there? */
               ASSIGN OPT=MDOCOUNTS VARS=&mdocounts,
                  FROM MDO=&mdo.datalist.data
               DO &cvar = 1 TO &mdocounts
                  SAY "&"&cvar" is "&mdo.datalist.data{&cvar}
                  END /*do*/
               END /*do*/
            OTHERWISE
               NOP
            END /*select*/
         END /*do*/
      ELSE
         SAY "This message does not have a USERMDO"
      MSGCONT
   END /*do*/
END zex1620n

This procedure is designed to loop forever to trap messages arriving at an OCS
window.  The MSGREAD verb specifies an MDO variable when reading a
message.  This allows you to read the whole message into the &$MSG. MDO
variable and, if the message contains an enclosed MDO variable, to automatically
read the enclosed MDO variable into the MDO variable specified by the
MSGREAD verb.  The message is analyzed to determine whether it contains an
enclosed MDO variable.  Depending on the map used to map the enclosed MDO
variable, information about the message is then displayed on the screen.

2. Install ZEX1620N as your active MSGPROC procedure, using the following
command from the OCS command input line:

PROFILE MSGPROC=ZEX1620N

3. Type the second NCL procedure as shown, calling it ZEX1621N:

zex1621n: PROCEDURE
   /* Sends text only */
   WRITE DATA=a b c
END zex1621n

4. Execute ZEX1621N from the OCS command input line by using the following
START command:

START ZEX1621N

5. Observe the results of installing ZEX1620N as your active MSGPROC procedure
and executing ZEX1621N, as shown in the following screen:

  (15:10) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1620N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   This message does not have a USERMDO
   This message does not have a USERMDO
   START ZEX1621N
   This message does not have a USERMDO
   a b c
   This message does not have a USERMDO
   NNM1005 START ZEX1621N PROCESSING COMPLETE. NCLID 002559
   This message does not have a USERMDO
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Sending Ordinary Variables You can send ordinary variables between NCL processes by using the ARGS or VARS
operand of the WRITE verb (with or without additional text).

When an INTREAD verb reads a message, the specified text is placed in the
&$INT.TEXT MDO variable.  If the DATA operand is not used or specifies a null
value, the value of &$INT.TEXT is null.  The ordinary variables, which are mapped by
the $NCL map, are placed in the &$INT.USERMDO MDO variable.  The name of the
map ($NCL) is placed in the &$INT.MAPNAME MDO variable.

The following WRITE verb sends ordinary variables, and also specifies the DATA
operand to send text:

…
&1 = x
&2 = y
&3 = z
WRITE NCLID=&nclid ARGS DATA=a b c
…

The preceding code segment has the same result as the following code segment:

…
&1 = x
&2 = y
&3 = z
ASSIGN MDO=&usermdo. MAP=$NCL FROM ARGS
WRITE NCLID=&nclid MDO=&usermdo. DATA=a b c
…

The following WRITE verb sends ordinary variables, but does not specify the DATA
operand and sends no text:

…
&1 = x
&2 = y
&3 = z
WRITE NCLID=&nclid ARGS
…

The preceding code segment has the same result as the following code segment:

…
&1 = x
&2 = y
&3 = z
ASSIGN MDO=&usermdo. MAP=$NCL FROM ARGS
WRITE NCLID=&nclid MDO=&usermdo.
…

The following table shows the result of using the WRITE verb to send ordinary
variables by using the ARGS or VARS operand (with and without the DATA operand
to send text), and the result of using an INTREAD verb that specifies PARSE=YES to
read the message:

WRITE Verb

INTREAD Verb

WRITE NCLID=&nclid
      ARGS
      DATA=a b c

WRITE NCLID=&nclid
      ARGS

INTREAD
PARSE=YES

&$INT.TEXT=a b c
&$INT.USERMDO=… x y z
&$INT.MAPNAME=$NCL

&$INT.TEXT=null
&$INT.USERMDO=… x y z
&$INT.MAPNAME=$NCL

INTREAD
PARSE=YES
ARGS

&$INT.TEXT=a b c
&1=x
&2=y
&3=z
&$INT.USERMDO=… x y z
&$INT.MAPNAME=$NCL

&$INT.TEXT=null
&1=x
&2=y
&3=z
&$INT.USERMDO=… x y z
&$INT.MAPNAME=$NCL

INTREAD
PARSE=YES
MDO=&mdo.

&$INT.TEXT=a b c
&$INT.USERMDO=… x y z
&$INT.MAPNAME=$NCL
&mdo.=… x y z

&$INT.TEXT=null
&$INT.USERMDO=… x y z
&$INT.MAPNAME=$NCL
&mdo.=… x y z

The following table shows the result of using the WRITE verb to send ordinary
variables by using the ARGS or VARS operand (with and without the DATA operand
to send text), and the result of using an INTREAD verb that specifies PARSE=NO to
read the message:

WRITE Verb

INTREAD Verb

WRITE NCLID=&nclid
      ARGS
      DATA=a b c

WRITE NCLID=&nclid
      ARGS

INTREAD
PARSE=NO

&$INT.TEXT=a b c
&$INT.USERMDO=… x y z
&$INT.MAPNAME=$NCL

&$INT.TEXT=null
&$INT.USERMDO=… x y z
&$INT.MAPNAME=$NCL

INTREAD
PARSE=NO
ARGS

&$INT.TEXT=a b c
&1=a b c
&2=null
&3=null
&$INT.USERMDO=… x y z
&$INT.MAPNAME=$NCL

&$INT.TEXT=null
&1=null
&2=null
&3=null
&$INT.USERMDO=… x y z
&$INT.MAPNAME=$NCL

INTREAD
PARSE=NO
MDO=&mdo.

&$INT.TEXT=a b c
&$INT.USERMDO=… x y z
&$INT.MAPNAME=$NCL
&mdo.=x y z

&$INT.TEXT=null
&$INT.USERMDO=… x y z
&$INT.MAPNAME=$NCL
&mdo.=x y z

Specifying Ordinary Variables When Reading a Message

The following NCL procedure uses the WRITE verb to send ordinary variables from
one NCL process to a target NCL process that specifies ordinary variables when
reading a message:

zex1618n: PROCEDURE
   /* Sends ordinary variables to an NCL process */
   /* Wait until ZEX1616N begins execution       */
   DO UNTIL &glblzex1616n \= ""
      DELAY 1
   END /*do*/
   &1 = x; &2 = y; &3 = z
   WRITE COLOR=WHITE,
         NRD=OPER,
         NCLID=&glblzex1616n,
         ARGS,
         DATA=a b c
END zex1618n

After following the steps outlined earlier in this section to execute ZEX1616N, type
ZEX1618N.  The following screen shows the results of executing ZEX1618N:

  (13:42) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1616N
   NCL ID of ZEX1616N is 270
   START ZEX1618N
   &1 is x
   &2 is y
   &3 is z
   This message has a USERMDO
   NNM1005 START ZEX1618N PROCESSING COMPLETE. NCLID 000271
   &$INT.TEXT is a b c
   &$INT.USERMDO is  . . .  x .  y .  z
   &$INT.MAPNAME is $NCL
   Elements and fields in &MDO. are
   Element DATALIST is  .  x .  y .  z
     Element DATA is x
     Element DATA{2} is y
     Element DATA{3} is z
   a b c
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Specifying an MDO Variable When Reading a Message

The following NCL procedure uses the WRITE verb to send ordinary variables from
one NCL process to another NCL process (MSGPROC) that specifies an MDO variable
when reading a message:

zex1622n: PROCEDURE
   /* Sends ordinary variables */
   /* mapped by $NCL, and text */
   &1 = x
   &2 = y
   &3 = z
   WRITE ARGS DATA=a b c
END zex1622n

After following the steps outlined earlier in this section to install ZEX1620N as your
active MSGPROC procedure, execute ZEX1622N to send a message to your OCS
window.  The following screen shows the results of executing the procedure:

  (15:12) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1620N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   This message does not have a USERMDO
   This message does not have a USERMDO
   START ZEX1622N
   This message does not have a USERMDO
   a b c
   This message has a USERMDO mapped by $NCL
   &MDO. =  . . .  x .  y .  z
   &1 is x
   &2 is y
   &3 is z
   NNM1005 START ZEX1622N PROCESSING COMPLETE. NCLID 002564
   This message does not have a USERMDO
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Sending MDO Variables You can send an MDO variable between NCL processes by using the MDO operand of
the WRITE verb (with or without additional text).  An MDO variable that is sent in a
message is called an enclosed MDO variable.  This is because the MDO variable, which
may be mapped by any map, is enclosed by the message, which travels in a data
structure that Mapping Services can interpret using the map $MSG:  the MDO variable
travels inside the message.

When an INTREAD verb reads a message containing an enclosed MDO variable, the
whole message is placed in an enclosing MDO variable.  (The &$INT. MDO variable
encloses another MDO variable.)  The specified text is placed in the &$INT.TEXT
MDO variable.  If the DATA operand is not used or specifies a null value, the value of
&$INT.TEXT is null.  The enclosed MDO variable, which is mapped by the map you
specify, is placed in the &$INT.USERMDO variable.  The name of the map you specify
is placed in the &$INT.MAPNAME MDO variable.

The following WRITE verb sends an enclosed MDO variable mapped by the map
$NCL, and also specifies the DATA operand to send text:

…
&1 = x
&2 = y
&3 = z
ASSIGN MDO=&usermdo. MAP=$NCL FROM ARGS
WRITE NCLID=&nclid MDO=&usermdo. DATA=a b c
…

The following WRITE verb sends an enclosed MDO variable mapped by the map
$NCL, but does not specify the DATA operand and sends no text:

…
&1 = x
&2 = y
&3 = z
ASSIGN MDO=&usermdo. MAP=$NCL FROM ARGS
WRITE NCLID=&nclid MDO=&usermdo.
…

The following WRITE verb sends an enclosed MDO variable mapped by the map
$MSG, and also specifies the DATA operand to send text:

…
ASSIGN MDO=&usermdo. MAP=$MSG
&usermdo.text = x y z
WRITE NCLID=&nclid MDO=&usermdo. DATA=a b c
…

The following WRITE verb sends an enclosed MDO variable mapped by the map
$MSG, but does not specify the DATA operand and sends no text:

…
ASSIGN MDO=&usermdo. MAP=$MSG
&usermdo.text = x y z
WRITE NCLID=&nclid MDO=&usermdo.
…

The following example uses the INTREAD verb to read a message, and then
immediately sends the message as an enclosed MDO variable to the NCL process with
the NCL ID of 1234:

…
INTREAD
WRITE NCLID=1234 MDO=&$INT. DATA=&$INT.TEXT
…

The WRITE verb in this example sends the text in &$INT.TEXT in two locations:  in the
enclosed MDO variable, &$INT., and as text specified by the DATA operand.  Sending
text in this way means that you do not have to extract the text from the enclosed MDO
variable when the message is read by an EMSREAD, INTREAD, or MSGREAD verb.

The following table shows the result of using the WRITE verb to send an enclosed
MDO variable by using the MDO operand (with and without the DATA operand to
send text), and the result of using an INTREAD verb that specifies PARSE=YES to read
the message:

WRITE Verb

INTREAD Verb

WRITE NCLID=&nclid
      MDO=&usermdo.
      DATA=a b c

WRITE NCLID=&nclid
      MDO=&usermdo.

INTREAD
PARSE=YES

&$INT.TEXT=a b c
&$INT.USERMDO=&usermdo
&$INT.MAPNAME=varies

&$INT.TEXT=null
&$INT.USERMDO=&usermdo
&$INT.MAPNAME=varies

INTREAD
PARSE=YES
ARGS

&$INT.TEXT=a b c
&$INT.USERMDO=&usermdo
&$INT.MAPNAME=varies

Mapped by $NCL
&1=x
&2=y
&3=z

Mapped by another map
&1=a
&2=b
&3=c

&$INT.TEXT=null
&$INT.USERMDO=&usermdo
&$INT.MAPNAME=varies

Mapped by $NCL
&1=x
&2=y
&3=z

Mapped by another map
&1=null
&2=null
&3=null

INTREAD
PARSE=YES
MDO=&mdo.

&$INT.TEXT=a b c
&$INT.USERMDO=&usermdo
&$INT.MAPNAME=varies
&mdo.=&usermdo

&$INT.TEXT=null
&$INT.USERMDO=&usermdo
&$INT.MAPNAME=varies
&mdo.=&usermdo

The following table shows the result of using the WRITE verb to send an enclosed
MDO variable by using the MDO operand (with and without the DATA operand to
send text), and the result of using an INTREAD verb that specifies PARSE=NO to read
the message:

WRITE Verb

INTREAD Verb WRITE NCLID=&nclid
      MDO=&usermdo.
      DATA=a b c

WRITE NCLID=&nclid
      MDO=&usermdo.

INTREAD
PARSE=NO

&$INT.TEXT=a b c
&$INT.USERMDO=&usermdo
&$INT.MAPNAME=varies

&$INT.TEXT=null
&$INT.USERMDO=&usermdo
&$INT.MAPNAME=varies

INTREAD
PARSE=NO
ARGS

&$INT.TEXT=a b c
&$INT.USERMDO=&usermdo
&$INT.MAPNAME=varies

Mapped by $NCL
&1=a b c
&2=null
&3=null

Mapped by another map
&1=a b c
&2=null
&3=null

&$INT.TEXT=null
&$INT.USERMDO=&usermdo
&$INT.MAPNAME=varies

Mapped by $NCL
&1=null
&2=null
&3=null

Mapped by another map
&1=null
&2=null
&3=null

INTREAD
PARSE=NO
MDO=&mdo.

&$INT.TEXT=a b c
&$INT.USERMDO=&usermdo
&$INT.MAPNAME=varies
&mdo.=&usermdo

&$INT.TEXT=null
&$INT.USERMDO=&usermdo
&$INT.MAPNAME=varies
&mdo.=&usermdo

Specifying Ordinary Variables When Reading a Message

The following NCL procedure uses the WRITE verb to send an MDO variable from
one NCL process to a target NCL process that specifies ordinary variables when
reading a message:

zex1619n: PROCEDURE
   /* Sends an MDO variable to an NCL process */
   /* Wait until ZEX1616N begins execution    */
   DO UNTIL &glblzex1616n \= ""
      DELAY 1
   END /*do*/
   ASSIGN MDO=&usermdo. MAP=$MSG
   &usermdo.text = x y z
   WRITE COLOR=WHITE,
         NRD=OPER,
         NCLID=&glblzex1616n,
         MDO=&usermdo.,
         DATA=a b c
END zex1619n

After following the steps outlined earlier in this section to execute ZEX1616N, type
ZEX1619N.  The following screen shows the results of executing ZEX1619N:

  (13:44) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1616N
   NCL ID of ZEX1616N is 272
   START ZEX1619N
   &1 is a
   &2 is b
   &3 is c
   This message has a USERMDO
   &$INT.TEXT is a b c
   &$INT.USERMDO is  . .x y z
   &$INT.MAPNAME is $MSG
   Elements and fields in &MDO. are
   Element TEXT is x y z
   a b c
   NNM1005 START ZEX1619N PROCESSING COMPLETE. NCLID 000273
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Specifying an MDO Variable When Reading a Message

The following NCL procedure uses the WRITE verb to send an enclosed MDO variable
from one NCL process to another NCL process (MSGPROC) that specifies an MDO
variable when reading a message:

zex1623n: PROCEDURE
   /* Sends an MDO variable mapped */
   /* by $MSG, and text            */
   ASSIGN MDO=&usermdo. MAP=$MSG
   &usermdo.text = x y z
   WRITE MDO=&usermdo. DATA=a b c
END zex1623n

After following the steps outlined earlier in this section to install ZEX1620N as your
active MSGPROC procedure, execute ZEX1623N to send a message to your OCS
window.  The following screen shows the results of executing the procedure:

  (15:13) --------------------- OPERATOR CONTROL SERVICES ----------------------
   PROFILE MSGPROC=ZEX1620N
   NNM0393 MSGPROC PROCESSING ACTIVATED
   NNM0357 PROFILE HAS CHANGED
   This message does not have a USERMDO
   This message does not have a USERMDO
   START ZEX1623N
   This message does not have a USERMDO
   a b c
   This message has a USERMDO mapped by $MSG
   &MDO.TEXT = x y z
   &1 is x
   &2 is y
   &3 is z
   NNM1005 START ZEX1623N PROCESSING COMPLETE. NCLID 002567
   This message does not have a USERMDO
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Sending a Message to
an NCL Process When

it Is Started

In addition to the START command, you can also use the START verb to invoke an
NCL procedure for asynchronous execution.  Unlike the START command, the START
verb can be executed only from an NCL process, not from a command input line.  The
START verb offers more flexibility than the START command when invoking an NCL
procedure.  The START verb is particularly well suited for use in NCL processes that
control the automation of operations tasks.

Specifying Variables When an NCL process is started, the START verb can send a message.  It can send the
message in one or more ordinary variables, specified by using the ARGS or VARS
operand, or in an MDO variable, which can be mapped by any map.  If an MDO
variable is sent, the entire contents, including the map header and the data, is placed in
the &$PRM. MDO variable in the started NCL process.  Additionally, the map used to
send the message must be used to interpret the message.  See Section 10, “Working
With Mapping Services,” for a description of MDO variables and maps.

For example, when an EMSPROC NCL procedure reads a message, the message is
placed in the &$EMS. MDO variable, which is mapped by the map $MSG.  The START
verb can send this message directly to a started NCL process.  In the started NCL
process, the same message is placed in the &$PRM. MDO variable, which is also
mapped by the map $MSG.

Specifying the Environment The START verb can specify the environment in which the started NCL process
executes.  By default, the started NCL process executes in the current execution
environment.  Other environments in which the started NCL process can execute are
the dependent processing environment associated with the invoking NCL process and
the background processing environments in which virtual users execute.

For example, you could start an NCL process used for the automation of an operations
task in the background processing environment in which BMON operates.  Executing
the NCL process in the BMON environment is useful because the NCL process
automatically sends the results of commands to all monitor-class users who are
profiled to receive them.  An example of an operations task is attempting to restart an
Expand or SNAX/XF line after EMSPROC receives a message that a line has failed.
The results could report whether the Expand or SNAX/XF line is successfully
restarted.

Synchronizing Execution The START verb can synchronize execution between the invoking NCL process and
the started NCL process by specifying NOTIFY=YES.  If execution is synchronized, the
invoking NCL process does not continue execution until it receives the results of
attempting to start the new NCL process.  If execution is successful, the NCL ID of the
started NCL process is available to the invoking NCL process in the &SYSMSG
variable and the &SYS.RETCODE system variable contains a return code of zero (0).  If
execution is unsuccessful, &SYSMSG is not affected and &SYS.RETCODE contains a
return code indicating an error.

If execution is not synchronized (NOTIFY=NO), the invoking NCL process continues
execution without waiting for the results of attempting to start the new NCL process.
Additionally, the NCL ID of the started NCL process is not made available to the
invoking NCL process and the value of &SYS.RETCODE is always 0.

For example, you could synchronize the execution of an NCL process used for the
automation of an operations task.  This is useful because it allows you to confirm that a
certain task has started.  If the task of attempting to restart an Expand or SNAX/XF
line is not started by the designated NCL process, then an error message could be
displayed.

Example The following NCL procedures show how to use the START verb to start an NCL
process.

The first NCL procedure is an EMSPROC procedure called ZEX1624N.  It is designed
to start one NCL process to automatically recover a failed Expand line and another
NCL process to automatically recover a failed SNAX/XF line.  Recovery is triggered
by specific event messages.  For more information on the EMSPROC NCL procedure,
see Section 17, “Developing System-Level NCL Procedures.”

The second and third NCL procedures (ZEX1625N and ZEX1626N), both of which are
started by the EMSPROC NCL procedure, perform the work of recovery.  ZEX1625N
attempts to recover an Expand line; ZEX1626N, a SNAX/XF line.  (By modifying the
code, you could use the same NCL process to perform both tasks.)  The information
required to attempt recovery is passed by the EMSPROC NCL procedure using the
START verb.

To see how these NCL procedures work, follow these steps:

1. Type the first NCL procedure as shown, calling it ZEX1624N:

zex1624n: PROCEDURE
   /* EMSPROC procedure for simple automation.  */
   /* An asynchronous NCL process is started by */
   /* the START verb to handle specific Expand  */
   /* and SNAX/XF messages, which report line   */
   /* problems.                                 */
   DO FOREVER
      EMSREAD VARS=&msgid
      IF &$ems.spi \= "" THEN DO
         SELECT &msgid
            /* Expand NET-LINE-NOT-READY  */
            WHEN "EXP0045" THEN DO
               START PROC=ZEX1625N ENV=BMON NOTIFY=YES MDO=&$ems.
               GOSUB CHECK_SUCCESS
               END /*do*/
            /* SNAX/XF LINE DOWN */
            WHEN "SX10023" THEN DO
               START PROC=ZEX1626N ENV=BMON NOTIFY=YES MDO=&$ems.
               GOSUB CHECK_SUCCESS
               END /*do*/
            /* SNAX/XF LINE INOP */
            WHEN "SX10021" THEN DO
               START PROC=ZEX1626N ENV=BMON NOTIFY=YES MDO=&$ems.
               GOSUB CHECK_SUCCESS
               END /*do*/
            OTHERWISE
               EMSCONT
         END /*select*/
      END /*do*/
   END /*do*/
   CHECK_SUCCESS:
      IF &sys.retcode = 0 THEN DO
         WRITE LOG=YES INTENS=HIGH,
            DATA=NCL PROCESS &sysmsg STARTED FOR EVENT &msgid
         EMSCONT INTENS=HIGH
         END /*do*/
      ELSE
         WRITE LOG=YES INTENS=HIGH,
            DATA=ERROR &sys.retcode STARTING NCL PROCESS
   RETSUB
END zex1624n

2. Copy ZEX1624N to the customized distribution library.

3. Install ZEX1624N as your EMSPROC procedure using the following SYSPARMS
command from the OCS command input line:

SYSPARMS EMSPROC=ZEX1624N

You can use the SHOW SYSPARMS command or the STATUS command to
confirm that ZEX1624N is now your active EMSPROC procedure.

4. Type the second NCL procedure as shown, calling it ZEX1625N:

zex1625n: PROCEDURE
   /* Started by EMSPROC for simple automation. */
   /* This asynchronous NCL process is started  */
   /* by the START verb to handle an Expand     */
   /* message, which reports line problems.     */
   CONTROL NOENDMSG
   &line = &$prm.spi.&($prm.spi.tandem.zems_tkn_subject_mark)
   INTCMD "OPSYS START EXPAND LINE "&line
   INTREAD
   /* You would normally add some checking here */
   /* to determine whether the line is started. */
   WRITE LOG=YES INTENS=HIGH,
      DATA="EXPAND LINE "&line" STARTED BY EMSPROC"
END zex1625n

5. Copy ZEX1625N to the user procedure library owned by BMON.  (You can
determine the name of this library by looking at the user ID definition record for
BMON in User ID Management Services (UMS).)

6. Type the third NCL procedure as shown, calling it ZEX1626N:

zex1626n: PROCEDURE
   /* Started by EMSPROC for simple automation. */
   /* This asynchronous NCL process is started  */
   /* by the START verb to handle a SNAX/XF     */
   /* message, which reports line problems.     */
   CONTROL NOENDMSG
   &line = &$prm.spi.&($prm.spi.tandem.zems_tkn_subject_mark)
   INTCMD "OPSYS START SNAX LINE "&line
   INTREAD
   /* You would normally add some checking here */
   /* to determine whether the line is started. */
   WRITE LOG=YES INTENS=HIGH,
      DATA="SNAX LINE "&line" STARTED BY EMSPROC"
END zex1626n

7. Copy ZEX1626N to the user procedure library owned by BMON.

8. Observe the results of starting an NCL process to recover an Expand line after
receiving an unsolicited message (EXP0045) reporting that an Expand line is not
ready:

  (12:26) --------------------- OPERATOR CONTROL SERVICES ----------------------
   12:25:51 EXP0045 LDEV 0027         NET: LINE NOT READY, ERROR #066
   12:25:54 E NCL PROCESS 109 STARTED FOR EVENT EXP0045
   12:25:55 EXP-003 \SYS1.$EXP02 LDEV 27 I/O ADDR: (1,0,%37,%2)
   12:25:55 EXP0117 LDEV 0027         NET: NETWORK REQUESTS ABORTED, CAUSE: 02
   12:25:58 M EXPAND LINE \SYS1.$EXP02 STARTED BY EMSPROC
   12:26:04 EXP0044 LDEV 0027         NET: LINE READY
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

First, the screen displays a message from EMSP, reporting an NCL process is
started to handle the event identified by message number EXP0045.  Second,
informational messages from the Expand subsystem are displayed.  Third, a
message is displayed from BMON, reporting the Expand line is started.  Finally, a
message is displayed reporting the Expand line is started; this message indicates
that automated recovery is successful.

Executing Persistent
NCL Processes

An NCL process is considered to be persistent when it has automatic restart capability.

Using the RESTART
Operand in the START Verb

When an NCL procedure is started by using the START verb with RESTART=YES or
RESTART=COND, and the START verb also specifies that the NCL process is started
in a background processing environment, a backup process is assigned by NonStop
NET/MASTER MS.  The backup process ensures that an NCL process automatically
restarts if a Guardian process or a hardware component fails.  An NCL process started
using this method is called a persistent NCL process.

When an NCL process restarts, the NCL ID of the restarted NCL process is unknown.
However, the restarted NCL process does not have the same NCL ID as the original
NCL process.

Number of Restarts in
Persistent NCL

To make an NCL process persistent, the NCL procedure must be started in a
background processing environment using the START verb, and the RESTART
operand must be equal to YES or COND.  The value of the &SYS.NCL.RESTART
system variable is initially set to 0 (zero) and increments by 1 each time the NCL
process restarts.

If you try to start the NCL procedure in other than a background processing
environment using the START verb with the RESTART=YES operand, then the
&SYS.RETCODE system variable is set to 48 and the NCL process does not start.

If you try to start the NCL procedure in other than a background processing
environment using the START verb with the RESTART=COND operand, then the
&SYS.RETCODE system variable is set to 52, and the NCL process is not persistent.  In
this case, the &SYS.NCL.RESTART system variable has a null value.

Persistent Versus Nonstop
NCL Processes

A persistent NCL process is not a nonstop NCL process.  (NonStop NET/MASTER MS
does not support nonstop NCL processes.)

If the CPU in which a persistent NCL process is executing fails for any reason, the
primary processing environment is dismantled and abends.  The primary processing
environment is not backed up.  Any file or message activity that has occurred during
the execution of the persistent NCL process is not rolled back when the NCL process is
restarted.  You can examine the activity log to find any messages generated by the
restarted NCL process.

For further information on persistent NCL, including the syntax of the START verb,
refer to the NonStop NET/MASTER NCL Reference Manual.


